
РОЗБІР ЗАВДАНЬ ІІ ЕТАПУ ВСЕУКРАЇНСЬКОЇ УЧНІВСЬКОЇ ОЛІМПІАДИ З ІНФОРМАТИКИ 2015-2016 НАВЧАЛЬНОГО РОКУ

8-11 КЛАС

Задача А: Школа 
Василь одержав з інформатики протягом семестру деякі бали і хоче дізнатися, на якому рівні він навчається. Рівень досягнень відповідає середньому балу (округленому до цілого за математичними правилами) за семестр: якщо бал від 1 до 3, то це початковий рівень (1), якщо від 4 до 6, то середній (2), якщо від 7 до 9, то достатній (3), якщо від 10 до 12, то високий (4).

Вхідні дані: 
У першому рядку дано число n (1≤n≤20) – кількість балів, що отримав  Василь за семестр. У другому дано n натуральних чисел – бали Василя (від 1 до 12).
Вихідні дані: 
Вивести одне з чотирьох чисел (1, 2, 3 або 4), що відповідають рівню навчальних досягнень Василя.

Приклад :
Вхідні дані:                                                          Вихідні дані: 

6                                                                     3
1 11 8 9 10 8

 Пояснення прикладу:       [image: image2.png]1+11+6+9+10+48

 (достатній рівень)
Розбір:

У даній задачі необхідно було знайти середнє арифметичне значення даних n чисел і визначити, до якого інтервалу це число належить. Але проблема виникає при середньому арифметичному значенні, що дорівнює 6,5. На мові програмування FreePascal функція round(6,5)=6, а не 7. При цьому round(9,5)=10, а round(3,5)=4, тобто округлення до парного цілого. Отже, краще не використовувати функцію round(). Зробимо математичні перетворення: 

 ( s<6,5n (10s<65n. Авторський розв’язок записаний нижче:

var a,s,n,i:longint;

begin

read(n);

s:=0;

for i:=1 to n do

 begin

  read(a);

  s:=s+a;

 end;

 if 10*s<35*n then write(1) else

 if 10*s<65*n then write(2) else

 if 10*s<95*n then write(3) else write(4);

 end.
Задача В: Банк 
Учень Петрик на уроці математики вивчав відсотки і розв’язував задачу про вклад у банку: «Вкладник поклав до банку a грн під p відсотків річних. Наприкінці кожного року з відсотків відкидається дробова частина копійок і додається до вкладу. Скільки років потрібно чекати, щоб вклад досяг b грн.?». Допоможіть Петрику розв’язати цю задачу.

Вхідні дані: 
У одному рядку дано три цілих числа a, p і b – початковий вклад, відсоток вкладу та необхідна сума (10≤a≤10000, a≤b≤10000, 1≤p≤100). 
Вихідні дані: 
Вивести одне натуральне число – кількість  років.

Приклад:

Вхідні дані:                                                  Вихідні дані: 

200 20 366                                                       4   
Пояснення прикладу. Через рік сума вкладу становитиме200+40=240 грн, через два роки – 288 грн, через три – 365,6 грн. Тому лише через чотири роки сума вкладу досягне необхідної суми.
Розбір:

Кількість років будемо знаходити за допомогою циклу while a<b do, тобто поки вклад а є меншим b. Відсотки будемо визначати за формулою trunc(a*p/100) і додавати до початкового вкладу. Але відкидається лише дробова частина копійок, тому потрібно було подати початковий і кінцевий вклад у копійках. Авторський розв’язок записаний нижче:

var a,b,p,k:longint;

begin

read(a,p,b);

a:=a*100; b:=b*100;

while a<b do

 begin

  a:=a+trunc(a*p/100);

  inc(k);

 end;

 write(k);

 end.
Задача С: Пірати (10 балів) 
У таємничому архіпелазі є n островів, кожен з яких задається двома координатами (х;у).  Одного дня на перший острів напали пірати і пограбували його. Кожного наступного дня вони захоплюють  найближчий острів (якщо таких виявляється більше одного, то захоплюють перший у списку). На який день пірати захоплять острів капітана Немо, що стоїть на k-му місці у списку островів?
Вхідні дані: 
У першому рядку дано натуральне число n(n≤1000) – кількість островів в архіпелазі. У наступних n рядках дано по два цілих числа x та y (-1000≤x,y≤1000) –  координати острова на площині.  В останньому рядку дано натуральне число k (1≤k≤n) – порядковий номер острова капітана Немо.
Вихідні дані: 
Вивести одне натуральне число – номер дня, коли буде захоплений острів капітана Немо.

Приклад:
Вхідні дані:                                                           Вихідні дані:  
5                                                                            4

2 6

1 -4

-3 2

5 1

-5 0
3
Пояснення прикладу.  На перший день пірати захоплять острів під номером 1, на другий – під номером 4, на третій – під номером 2, а на четвертий – острів 3 (відстань від другого острова до третього і п’ятого однакова, але острів 3 стоїть раніше в списку).

Розбір:

Задамо початкові умови: 

k=1(кількість островів, що вже захоплені),

j=1 (порядковий номер острова, що захоплений)

f[j]=1 (прапорець, який показує, що острів j вже захоплений)

Задамо цикл while j<>m do (поки не захоплений острів капітана Немо m повторюємо)
У циклі на кожному кроці будемо повторювати такі дії:

1. збільшувати k;

2. min=maxlongint (початкова мінімальна відстань  – є найбільше число типу longint);

3. Пробігаємо список усіх островів і шукаємо серед непомічених островів (f[i]=0) той, до якого відстань є мінімальною і запам’ятовуємо його номер p;

4. j=p;

5. f[j]=1.

Відповіддю на задачу буде число k. Авторський розв’язок записаний нижче:

var k,i,n,m,j,p,min:longint;

x,y,f:array[1..1000] of longint;

begin
read(n);

for i:=1 to n do

 read(x[i],y[i]);

read(m);

k:=1;

j:=1;

f[j]:=1;

while j<>m do

 begin

  inc(k);

  min:=maxlongint;

  for  i:=1 to n do

    if (sqr(x[j]-x[i])+sqr(y[j]-y[i])<min) and (f[i]=0) then

                            begin

                             min:=sqr(x[j]-x[i])+sqr(y[j]-y[i]);

                             p:=i;

                            end;

   j:=p;

   f[j]:=1;

 end;

 write(k);

 end.

Задача D: Дроби 
Дано два дроби у вигляді [image: image5.png]


 і [image: image7.png][



. Знайти їх суму і подати у вигляді нескоротного дробу[image: image9.png]


. (A,D,X – ціла частина дробу, B, E, Y – чисельник, C, F, Z – знаменник). Якщо дробової частини дріб не містить, то вивести тільки Х.
Вхідні дані: 
Дано шість цілих чисел A, B, C, D, E, F (0≤ A,B,C,D,E,F≤1000, С≠0, F≠0).
Вихідні дані: 
Вивести три цілих числа X, Y, Z, якщо Y≠0, або одне число X, якщо Y=0.
Приклад  1:
Вхідні дані:                                                Вихідні дані: 
2 2 3 3 5 6                                               6 1 2     

Приклад  2:
Вхідні дані:                                               Вихідні дані: 
10 1 4 0 3 4                                             11 
Пояснення прикладу 1. [image: image11.png]


    

Пояснення прикладу 2. [image: image13.png]10-+2=102=11

Розбір:

Зробимо математичні перетворення: 
[image: image14.wmf](

Позначимо чисельник дробу змінною ch, а знаменник – zn. Щоб подати його у вигляді нескоротного дробу цей дріб потрібно скоротити, тобто поділити на їх НСД (найбільший спільний дільник), який знаходимо рекурсивною функцією за алгоритмом Евкліда. Після скорочення дробу виділяємо цілу частину X і остачу Y. Виводимо на екран згідно умови. Авторський розв’язок записаний нижче:

var ch,zn,a,b,c,d,e,f,p:longint;

function nsd(x,y:longint):longint;

begin

 if y>0 then nsd:=nsd(y,x mod y) else nsd:=x;

end;

begin

read(a,b,c,d,e,f);

ch:=(a*c+b)*f+(d*f+e)*c;

zn:=c*f;

p:=nsd(ch,zn);

ch:=ch div p;

zn:=zn div p;

if ch mod zn=0 then write(ch div zn) else

write(ch div zn,' ',ch mod zn,' ',zn);

 end.

Задача E: «Лицар» 
	
	
	
	
	

	
	
	
	
	

	
	
	
	
	


У давньому королівстві лицарю за добру службу дали можливість пройти через королівську скарбницю і забрати стільки золотих монет, скільки є в тих кімнатах, через які він буде проходити. Лицар добув план скарбниці і встановив, що вона складається з n рядів по m кімнат у кожному. З кожної кімнати ведуть троє дверей до кімнат нижнього ряду (нижньої і двох сусідніх до неї). Кожні двері відчиняються тільки в одному напрямі, тобто повернутися назад неможливо. Лицар може почати рух з будь-якої верхньої кімнати (першого ряду). Відомо, скільки монет знаходиться в кожній кімнаті. Яку максимальну кількість монет може винести лицар із скарбниці?
Вхідні дані: 
У першому рядку дано два цілих числа n і m( 1≤n, m≤100) – розміри скарбниці. У наступних n рядках дано m цілих чисел a[i,j] – кількість монет, що знаходиться у кімнаті у і-му рядку і
 j-му стовпці (0≤a[I,j]≤100).
Вихідні дані:
Максимальну кількість монет, що може винести із скарбниці лицар.
Приклад :

Вхідні дані:                                                              Вихідні дані: 
4 5                                                                          24

2 1 5 2 7

2 4 5 1 2

9 2 4 1 0

6 5 4 3 2

Пояснення прикладу.
	2
	1
	5
	2
	7

	2
	4
	5
	1
	2

	9
	2
	4
	1
	0

	6
	5
	4
	3
	2


Розбір:

У кожну кімнату х[i,j] можна потрапити лише з комірок [i-1,j-1], [i-1,j] та [i-1,j+1]. Щоб отримати максимальну суму потрібно для кожної кімнати, куди потрапить лицар знаходити максимальне значення з трьох верхніх кімнат (використовуємо функцію max). Ці дії слід виконати для кожного рядка, починаючи з другого і закінчуючи останнім. З останнього рядка вибрати максимальне значення max1 – це і буде відповіддю на задачу. Так як для першої та останньої клітинки кожного рядка ми будемо звертатися до нульової та (m+1)-ої, то при описі масиву потрібно це врахувати a:array[1..100,0..101] of longint. Авторський розв’язок записаний нижче:

var n,m,i,j,max1:longint;

a:array[1..100,0..101] of longint;

function max(x,y:longint):longint;

begin

 if x>y then max:=x else max:=y;;

end;

begin

read(n,m);

for i:=1 to n do

 for j:=1 to m do

  read(a[i,j]);

for i:=2 to n do

  for j:=1 to m do

   a[i,j]:=a[i,j]+max(max(a[i-1,j-1],a[i-1,j]),a[i-1,j+1]);

 for j:=1 to m do

  if a[n,j]>max1 then max1:=a[n,j];

  write(max1);

end.

